query: SELECT ?competition ?competitionLabel ?date ?end WHERE {
    { ?competition wdt:P31 wd:Q2990963. }
    UNION
    {
      ?competition wdt:P31 wd:Q16510064.
      ?competition wdt:P361 ?comp.
      ?comp wdt:P31 wd:Q2990963.
    }
    UNION
    {
      ?competition wdt:P361 ?segment.
      ?segment wdt:P31 wd:Q16510064.
    }
    OPTIONAL { ?competition wdt:P580 ?start. }
    OPTIONAL { ?competition wdt:P582 ?end. }
    OPTIONAL { ?competition wdt:P585 ?pointInTime. }
    BIND(COALESCE(?start, ?pointInTime) AS ?date)
    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
    FILTER((BOUND(?start)) || (BOUND(?pointInTime)))
    FILTER(?end >= (NOW() - "P1D"^^xsd:duration))
    FILTER((NOW()) >= ?date)
  }
  ORDER BY ?date